
Пример1. Работа с текстовыми данными и массивами

unit sort2_;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls;

type

  TForm1 = class(TForm)

    Label1: TLabel;

    Edit1: TEdit;

    Label2: TLabel;

    Button1: TButton;

    Label3: TLabel;

    procedure Button1Click(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation

{$R *.DFM}

{ Функция возвращает подстроку с указанным номером }

Function GetSubStr3(st:string;     { строка }

                  n:integer)       { номер подстроки }

                  :string;         { подстрока или '' }

var

   p:integer;

   i:integer;

begin

   For i:=1 to n-1 do

   begin

     p:=pos(' ',st);

     if p=0 then begin

      GetSubStr3:=''; exit;

     end;

     st:=copy(st,p+1,Length(st)-p);

     //если в начале оставшейся части строки

     //есть пробелы, то удалим их }

     while (pos(' ',st)=1)and (length(st)>0) do

         delete(st,1,1);

   end;

   p:=pos(' ',st);

   if  p <> 0

      then result:=copy(st,1,p-1)

      else result:=st;

end;

procedure TForm1.Button1Click(Sender: TObject);

const

   SIZE=5;

var

   a:array[1..SIZE] of integer;

   k:integer;       { текущий элемент массива }

   i:integer;       { индес для ввода и вывода массива }

   changed:boolean; { TRUE, если в текущем цикле были обмены }

   buf:integer;     { буфер для обмена элемнтов массива }

begin

   { ввод массива }

   for i:=1 to SIZE do

       a[i]:=StrToInt(GetSubstr3(Edit1.text,i));

   label2.caption:='';

   { сортирвка массива }

    repeat

       changed:=FALSE;

       for k:=1 to SIZE-1 do

         begin

           if a[k] > a[k+1]

             then begin

                { обменяем k-ый и k+1-ый элементы }


        buf:=a[k];


        a[k]:=a[k+1];


        a[k+1]:=buf;

                changed:=TRUE;

             end;

         end;

         { вывод массива }


 for i:=1 to SIZE do

             Label2.caption:=label2.caption+' '+IntTostr(a[i]);

         Label2.caption:=label2.caption+#13;

    until not changed;

    Label2.caption:=label2.caption+#13+'Массив отсортирован.';

end;

end.
Пример2. Работа со списком и переключателем

unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls;

type

  TForm1 = class(TForm)

    Label2: TLabel;

    Edit1: TEdit;

    Button1: TButton;

    Label1: TLabel;

    Label3: TLabel;

    ListBox1: TListBox;

    Label4: TLabel;

    Label5: TLabel;

    procedure FormCreate(Sender: TObject);

    procedure Button1Click(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation

{$R *.DFM}

procedure TForm1.FormCreate(Sender: TObject);

begin

ListBox1.items.add('Россия');

ListBox1.items.add('Австрия');

ListBox1.items.add('Англия');

ListBox1.items.add('Германия');

ListBox1.items.add('Дания');

ListBox1.items.add('Исландия');

ListBox1.items.add('Италия');

ListBox1.items.add('Нидерланды');

ListBox1.itemindex:=0;

end;

procedure TForm1.Button1Click(Sender: TObject);

var

   funt:real; { вес в фунтах }

   kg:real;   { вес в килограммах }

   k:real;    { коэффициент пересчета }

begin

 case ListBox1.ItemIndex of

      0: k:=0.4059;  { Россия }

      1: k:=0.453592;{ Англия }

      2: k:=0.56001; { Австрия }

      3..5,7:k:=0.5; { Германия, Дания, Исландия, Нидерланды }

      6: k:=0.31762; { Италия }

      else Label5.Caption:='Ошибка оператора Case!';

 end;

 funt:=StrToFloat(Edit1.Text);

 kg:=k*funt;

 label4.caption:=Edit1.Text

                 + ' фунт(а/ов) - это '

                 + FloatToStrF(kg,ffFixed,6,3)

                 + 'кг.';

end;

end.
Пример 3. Диалоги и связанные формы. Текстовый редактор файлов *.rtf с любым количеством окон

Модуль MDIFrame.pas
{Форма для открытия редактора}

unit MDIFrame;

interface

uses

  SysUtils, Windows, Messages, Classes, Graphics, Controls, Forms, Dialogs,

  Menus;

type

  TFrameForm = class(TForm)

    MainMenu1: TMainMenu;

    File1: TMenuItem;

    New1: TMenuItem;

    Open1: TMenuItem;

    N1: TMenuItem;

    Exit1: TMenuItem;

    Window1: TMenuItem;

    Tile1: TMenuItem;

    Cascade1: TMenuItem;

    OpenFileDialog: TOpenDialog;

    procedure Exit1Click(Sender: TObject);

    procedure New1Click(Sender: TObject);

    procedure Tile1Click(Sender: TObject);

    procedure Cascade1Click(Sender: TObject);

    procedure Open1Click(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  FrameForm: TFrameForm;

implementation

uses MDIEdit; //Второй модуль проекта

{$R *.DFM}

procedure TFrameForm.Exit1Click(Sender: TObject);

begin {Меню Файл пункт Выход}

  Close;

end;

procedure TFrameForm.New1Click(Sender: TObject);

begin {Меню Файл пункт Новый  - создаем форму редактирования,

которая является дочерней формой }

  TEditForm.Create(Self);

end;

procedure TFrameForm.Tile1Click(Sender: TObject);

begin {Метод укладывает окна дочерней формы мозаикой}

  Tile;

end;

procedure TFrameForm.Cascade1Click(Sender: TObject);

begin {Метод укладывает окна дочерней формы каскадом}

  Cascade;

end;

procedure TFrameForm.Open1Click(Sender: TObject);

begin {если выполнен диалог Открыть}

  if OpenFileDialog.Execute then

  with TEditForm.Create(Self) do   {Создав форму редактора,}

    Open(OpenFileDialog.FileName); {выполнить с ней метод открытия,

    передав имя файла, полученное из диалога}

end;

end.
Модуль MDIEdit.pas

{Форма редактора}

unit MDIEdit;

interface

uses

  SysUtils, Windows, Messages, Classes, Graphics, Controls, Forms, Dialogs,

  Menus, StdCtrls, ComCtrls;

type

  TEditForm = class(TForm)

    MainMenu1: TMainMenu;

    File1: TMenuItem;

    New1: TMenuItem;

    Open1: TMenuItem;

    Save1: TMenuItem;

    Saveas1: TMenuItem;

    Print1: TMenuItem;

    N2: TMenuItem;

    Exit1: TMenuItem;

    Edit1: TMenuItem;

    Cut1: TMenuItem;

    Copy1: TMenuItem;

    Paste1: TMenuItem;

    Delete1: TMenuItem;

    N3: TMenuItem;

    Selectall1: TMenuItem;

    Character1: TMenuItem;

    Left1: TMenuItem;

    Right1: TMenuItem;

    Center1: TMenuItem;

    N4: TMenuItem;

    Wordwrap1: TMenuItem;

    N5: TMenuItem;

    Font1: TMenuItem;

    Editor: TRichEdit; {Объект TRichEdit - это редактор текста}

    PopupMenu1: TPopupMenu;

    Cut2: TMenuItem;

    Copy2: TMenuItem;

    Paste2: TMenuItem;

    SaveFileDialog: TSaveDialog;

    FontDialog1: TFontDialog;

    Printersetup1: TMenuItem;

    Close1: TMenuItem;

    PrinterSetupDialog1: TPrinterSetupDialog;

    PrintDialog1: TPrintDialog;

    N1: TMenuItem;

    FindDialog1: TFindDialog;

    N6: TMenuItem;

    Find1: TMenuItem;

    procedure Exit1Click(Sender: TObject);

    procedure New1Click(Sender: TObject);

    procedure AlignClick(Sender: TObject);

    procedure Wordwrap1Click(Sender: TObject);

    procedure Cut1Click(Sender: TObject);

    procedure Copy1Click(Sender: TObject);

    procedure Paste1Click(Sender: TObject);

    procedure Selectall1Click(Sender: TObject);

    procedure Delete1Click(Sender: TObject);

    procedure Edit1Click(Sender: TObject);

    procedure Saveas1Click(Sender: TObject);

    procedure Save1Click(Sender: TObject);

    procedure Font1Click(Sender: TObject);

    procedure Close1Click(Sender: TObject);

    procedure FormClose(Sender: TObject; var Action: TCloseAction);

    procedure FormCloseQuery(Sender: TObject; var CanClose: Boolean);

    procedure FormCreate(Sender: TObject);

    procedure Printersetup1Click(Sender: TObject);

    procedure Print1Click(Sender: TObject);

    procedure Open1Click(Sender: TObject);

    procedure FindDialog1Find(Sender: TObject);

    procedure Find1Click(Sender: TObject);

  private

    { Private declarations }

    PathName: string;

  public

    { Public declarations }

    procedure Open(const AFileName: string);

  end;

var

  EditForm: TEditForm;

const {Имя файла по умолчанию}

  DefaultFileName = 'Без имени.rtf';

implementation

uses Clipbrd, //Библиотека буфера обмена
     Printers, //Библиотека принтера

     MDIFrame; //Второй модуль проекта

{$R *.DFM}

procedure TEditForm.Exit1Click(Sender: TObject);

begin {Меню Файл, пункт Выход - обращаемся к методу формы FrameForm}

  FrameForm.Exit1Click(Sender);

end;

procedure TEditForm.New1Click(Sender: TObject);

begin {Меню Файл, пункт Новый - обращаемся к методу формы FrameForm}

  FrameForm.New1Click(Sender);

end;

procedure TEditForm.Open1Click(Sender: TObject);

begin {Меню Файл, пункт Открыть - обращаемся к методу формы FrameForm}

  FrameForm.Open1Click(Sender);

end;

procedure TEditForm.AlignClick(Sender: TObject);

begin  { У пунктов меню Текст Левый край, Правый край, Центр -

 один обработчик }

  Left1.Checked := False;   {Сбрасываем все флажки}

  Right1.Checked := False;

  Center1.Checked := False;

  with Sender as TMenuItem do Checked := True;

  {Прием для установки свойства}

  with Editor.Paragraph do {С текущим параграфом выполнить}

    {В зависимоcти от выбранного флажка}

    if Left1.Checked then

      Alignment := taLeftJustify {по левому краю}

    else if Right1.Checked then

      Alignment := taRightJustify {по правому краю}

    else if Center1.Checked then

      Alignment := taCenter; {по центру}

end;

procedure TEditForm.Wordwrap1Click(Sender: TObject);

begin {Меню Текст, п. Перенос по словам}

  with Editor do

  begin

    WordWrap := not WordWrap; { Переключили свойство выравнивания по словам у объекта типа TRichEdit }

    if WordWrap=true then {Если включен перенос, то только вертикальная полоса выравнивания}

      ScrollBars := ssVertical

    else

      ScrollBars := ssBoth; {Иначе обе полосы}

    WordWrap1.Checked := WordWrap; { Запомнили состояние свойства выравнивания в элементе меню }

  end;

end;

procedure TEditForm.Cut1Click(Sender: TObject);

begin {Меню Правка, Пункт Вырезать - вызываем метод Вырезать}

  Editor.CutToClipboard;

end;

procedure TEditForm.Copy1Click(Sender: TObject);

begin {Меню Правка, Пункт Копировать - вызываем метод Копировать}

  Editor.CopyToClipboard;

end;

procedure TEditForm.Paste1Click(Sender: TObject);

begin {Меню Правка, Пункт Вставить - вызываем метод Вставить}

  Editor.PasteFromClipboard;

end;

procedure TEditForm.Selectall1Click(Sender: TObject);

begin {Меню Правка, Пункт Выделить все - вызываем метод Выделить все}

  Editor.SelectAll;

end;

procedure TEditForm.Delete1Click(Sender: TObject);

begin {Меню Правка, Пункт Удалить - вызываем метод Удалить}

  Editor.ClearSelection;

end;

procedure TEditForm.Edit1Click(Sender: TObject);

var

  HasSelection: Boolean;

begin

  Paste1.Enabled := Clipboard.HasFormat(CF_TEXT);

  Paste2.Enabled := Paste1.Enabled;

  HasSelection := Editor.SelLength > 0;

  Cut1.Enabled := HasSelection;

  Cut2.Enabled := HasSelection;

  Copy1.Enabled := HasSelection;

  Copy2.Enabled := HasSelection;

  Delete1.Enabled := HasSelection;

end;

procedure TEditForm.Open(const AFileName: string);

begin {Функция открытия файла, которую мы сделали членом класса}

  PathName := AFileName;

  Caption := ExtractFileName(AFileName);

   {сделать заголовок окна текста = имя файла}

  with Editor do begin

    Lines.LoadFromFile(PathName); {Загрузить текст из файла}

    SelStart := 0; {НАчало выбранного фрагмента}

    Modified := False; {Флаг "файл был изменен"}

  end;

end;

procedure TEditForm.Saveas1Click(Sender: TObject);

begin {Меню Файл, пункт Сохранить как}

  SaveFileDialog.FileName := PathName;

  {Теперь нам пригодилось то, что имя открываемого файла мы сохранили

  в строке PathName - присваиваем ее диалогу Сохранить как...}

  if SaveFileDialog.Execute then {Есди диалог сохранения выполнен,}

  begin

    PathName := SaveFileDialog.FileName; {Снова запомнить имя}

    if Pos('.rtf',PathName)=0 then PathName:=PathName+'.rtf';

    Caption := ExtractFileName(PathName); {И вывести новое имя в заголовке окна}

    Save1Click(Sender); {Вызвать функцию пункта Сохранить}

  end;

end;

procedure TEditForm.Save1Click(Sender: TObject);

begin {Меню Файл, пункт Сохранить}

  if PathName = DefaultFileName then

   {Если имя по умолчанию - то вызвать функцию пункта Сохранить как...}

    SaveAs1Click(Sender)

  else

  begin

    Editor.Lines.SaveToFile(PathName); {Метод Сохранения в файл компоненты RichEdit}

    Editor.Modified := False; {Сняли флажок "Изменен" компонента -

    т.к. флажок ставится автоматически при внесении изменений в текст}

  end;

end;

procedure TEditForm.Font1Click(Sender: TObject);

begin {Меню Текст, пункт Шрифт...}

  FontDialog1.Font := Editor.Font;

   {В качестве выбранного в диалоге шрифта взять текущий}

  if FontDialog1.Execute then {Если диалог выбора шрифта выполнен}

    Editor.SelAttributes.Assign(FontDialog1.Font);

     {Атрибуты текста компонента взять из диалога!}

end;

procedure TEditForm.Close1Click(Sender: TObject);

begin {Меню Файл, пункт Выход}

  Close;

end;

procedure TEditForm.FormClose(Sender: TObject; var Action: TCloseAction);

begin {Метод, созданный на закрытие окна формы}

  Action := caFree; {Закрыть форму}

{  Другие возможности:

 caNone       - ничего не делать

 caHide      - скрыть окно формы

 caMinimize) - минимизировать окно формы }

end;

procedure TEditForm.FormCloseQuery(Sender: TObject; var CanClose: Boolean);

const {Метод, созданный на запрос закрытия окна формы}

  SWarningText = 'Сохранить изменения в файле "%s"?';

begin

  if Editor.Modified then {Если были изменения}

  begin

    case MessageDlg(Format(SWarningText, [PathName]), mtConfirmation,

      [mbYes, mbNo, mbCancel], 0) of {Вывести окно диалога}

      idYes:    Save1Click(Self); {Выбор Да - вызов метода Сохранить}

      idCancel: CanClose := False; {Выбор Нет - снять флажок "Можно закрывать"}

    end;

  end;

end;

procedure TEditForm.FormCreate(Sender: TObject);

begin {Обработчик создания формы - установить имя файла по умолчанию}

  PathName := DefaultFileName;

end;

procedure TEditForm.Printersetup1Click(Sender: TObject);

begin {Меню Файл, пункт Установки принтера...}

  PrinterSetupDialog1.Execute;

end;

procedure TEditForm.Print1Click(Sender: TObject);

begin {Меню Файл, пункт Печать}

  if PrintDialog1.Execute then

    Editor.Print(PathName);

end;

procedure TEditForm.FindDialog1Find(Sender: TObject);

var

  FoundAt: LongInt;

  StartPos, ToEnd: integer;

begin

  with Editor do begin

    StartPos := 0;

    ToEnd := Length(Text);

    FoundAt := FindText(FindDialog1.FindText, StartPos, ToEnd,

     [stMatchCase]);

    if FoundAt <> -1 then begin

      SetFocus;

      SelStart := FoundAt;

      StartPos := SelStart;

      ScrollBy (0,StartPos);

      SelLength := Length(FindDialog1.FindText);

    end;

  end;

end;

procedure TEditForm.Find1Click(Sender: TObject);

begin

  FindDialog1.Execute;

end;

end.

Пример 4. Рисуем графику на форме

[image: image5.png]"~

T

aren.





unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  ExtCtrls, StdCtrls;

type

  TForm1 = class(TForm)

    Refresh: TButton;

    DrawButton: TButton;

    procedure FormPaint(Sender: TObject);

    procedure RefreshClick(Sender: TObject);

    procedure DrawButtonClick(Sender: TObject);

  private

    { Private declarations }

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation

const StrokIcons: array [0..4] of PChar = ( //Стандартные иконки из Windows

 IDI_APPLICATION,

 IDI_ASTERISK,

 IDI_EXCLAMATION,

 IDI_HAND,

 IDI_QUESTION

);

var IconList: TList; //Объект типа Список
{$R *.DFM}

procedure TForm1.FormPaint(Sender: TObject);

var Icon:TIcon; //Переменная типа Иконка

I,sx1,sy1,sx2,sy2,width,height:Integer;

s:string;

R: TRect;

begin

 IconList:=TList.Create; //Создали список
 For i:=Low(StrokIcons) to High(StrokIcons) do begin

  Icon:=TIcon.Create; //Создаем очередную иконку

  Icon.Handle:=LoadIcon(0,Strokicons[i]); //И добавляем ее
  IconList.Add(TIcon(Icon));

 end;

 with Canvas do begin

  Brush.Color:=clYellow;

  //Цвет заливки
  Brush.Style:=bsBdiagonal;

  //Другие стили

  // сплошеня - bsSolid

  // нет заливки - bsClear

  // bsFDiagonal

  Pen.Color:=clGreen;

  Rectangle (ClipRect.Left,ClipRect.Top,ClipRect.Right,ClipRect.Bottom);

  //Прямоугольник на всю форму
  For i:=Low(StrokIcons) to High(StrokIcons) do begin

   Draw (I*40,10,TICon(Iconlist[i]));

  end; //вывели массив иконок

  //Управляем шрифтом:

  Font.Color:=clBlue;

  Font.Name:='Arial'; //Имя шрифта. Можно использовать

  //Times New Roman  - основной текст

  //Courier - все символы одинаковой ширины

  //Arial - заголовки

  Font.Size:=20;

  Font.Style:=[fsItalic,fsBold];//Курсив и жирный

  s:='Hello';

  width:=TextWidth(s);

  height:=TextHeight(s);

  TextOut (100,100,s);

  Font.Size:=10;

  Font.Style:=[];

  TextOut (100,130,'ширина='+IntToStr(width)+' высота='+IntToStr(height));

  //Используем метод GetClientRect:

  R := GetClientRect;

  //Есть и другие методы определения размеров клиентской области окна -

  // GetClientOrigin, ClientHeight, ClientWidth

  Arc(R.Left, R.Top, R.Right, R.Bottom, R.Left, R.Top, R.Right, R.Top);

{ Первые 4 параметра - размеры прямоугольника, в который вписана дуга

  Остальные:

  R.Left, R.Top, - начало дуги

  R.Right, R.Top - конец дуги

  рисуется против часовой стрелки от начала к концу

}

  Chord(R.Left, R.Top, R.Right, R.Bottom, R.Left, R.Top, R.Right, R.Top);

 //Соединили отрезком концы дуги
  for i:=100 to 200 do Pixels[i,100]:=clRed;

  // Pixels[x-координата, у-коррдината] - доступ к отдельным точкам

  // возвращает тип TColor

 end;

 //Рисуем случайные прямоугольники:

 Randomize; //сгенерировали 1-е случайное число

 Canvas.Brush.Style:=bsClear; //Прямоугольники не должны быть заполненными

 for i:=1 to 5 do begin

  sy1:=random (R.Bottom-R.Top);

  sx1:=random (R.Right-R.Left);

  sx2:=sx1+random (R.Right-sx1);

  sy2:=sy1+random (R.Bottom-sy1);

  Canvas.Rectangle (sx1,sy1,sx2,sy2);

 end;

end;

procedure TForm1.RefreshClick(Sender: TObject);

var R: TRect; //Кнопка Очистить
begin

 with Canvas do begin

  Brush.Style:=bsSolid;

  R := GetClientRect;

  Pen.Color:=clWhite;

  rectangle(R.Left, R.Top, R.Right, R.Bottom);

  Refresh;

 end;

end;

procedure TForm1.DrawButtonClick(Sender: TObject);

begin //Кнопка Заново
 FormPaint(Form1);

end;

end.

Пример 5. Работаем с цветами

unit Colors;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  ExtCtrls, StdCtrls;

type

  TForm1 = class(TForm)

    ListBox1: TListBox;

    Button1: TButton;

    Panel1: TPanel;

    Label1: TLabel;

    Label2: TLabel;

    procedure Button1Click(Sender: TObject);

    procedure ListBox1Click(Sender: TObject);

  private

    { Private declarations }

    procedure GetOneColor (const S:String);

  public

    { Public declarations }

  end;

var

  Form1: TForm1;

implementation

{$R *.DFM}

procedure TForm1.GetOneColor(Const S:String);

begin

 ListBox1.Items.Add(S);

end;

procedure TForm1.Button1Click(Sender: TObject);

begin

 GetColorValues (GetOneColor);

end;

procedure TForm1.ListBox1Click(Sender: TObject);

begin

 with ListBox1 do begin

  Panel1.Caption:=Items[ItemIndex];

  Panel1.Color:=StringToColor(Items[ItemIndex]);

  Label1.Caption:=IntToStr(ColorToRGB(StringToColor(Items[ItemIndex])));

 end;

end;

end.
